Pensieve header: Implementing ordering symbols for U(g0).

Reminders



$R=\sum a_i\otimes b_i \in A\otimes A = \hat{U}(\mathfrak{g})\otimes \hat{U}(\mathfrak{g})$

s.t. $R^{12}R^{13}R^{23}=R^{23}R^{13}R^{12}$

$\to \sum_{i,j,k} b_i a_j b_k a_i b_j a_k \in U(\mathfrak{g})$



PBW: $\mathfrak{g}=\langle x_1,\dots x_k\rangle \Rightarrow \{x_1^{a_1}x_2^{a_2}\cdots x_k^{a_k} : a_i\in\mathbb{Z}_{\geq 0}\}$ is a basis of $U(\mathfrak{g})$.

Today: $\mathfrak{g}_0=\langle h,e,l,f\rangle / \begin{matrix} h \text{ central} \\ [e,l]=-l \quad [f,l]=f \quad [e,f]=h\end{matrix}$

$r=h\otimes l+e\otimes f \qquad R=\exp(r)$

Note $U(\mathfrak{g}_0)^{\otimes S}=U(\oplus_S \mathfrak{g}_0)=U\left(\langle h_i,e_i,l_i,f_i\rangle / \begin{matrix} h_i \text{ central} \\ [e_i,l_j]=\delta_{ij}\,l_i \text{ etc.}\end{matrix}\right)$

Implementing g_0

```
PBWRule = {e → 1, l → 2, f → 3};
B[U@e, U@l] = -U@e;
B[U@f, U@l] = U@f;
B[U@e, U@f] = h U[];
```

```
$TD = 3;
ℏ /: ℏ^d_. /; d > $TD := 0;
```

```
ℏ^d_. /; d > $TD := 0;
```

SetDelayed: Tag Power in ℏ^d_. /; d > $TD is Protected.

```
x_ ≤ y_ := OrderedQ[{x, y} /. PBWRule];
x_ ≺ y_ := !OrderedQ[{y, x} /. PBWRule];
Simp[ℰ_] := Collect[ℰ, _U, Expand];
```

```
U_i_[ℰ_] := ℰ /. {h → h_i, t → t_i, u_U :→ Replace[u, x_ :→ x_i, 1]};
B[U[(x_)_i_], U[(y_)_i_]] := B[U[x_i], U[y_i]] = U_i[B[U@x, U@y]];
B[U[(x_)_i_], U[(y_)_j_]] /; i =!= j := 0;
B[x_, x_] = 0;
B[U[y_], U[x_]] := B[U[y], U[x]] = Simp[-B[U[x], U[y]]];
B[x_, y_] := x ** y - y ** x;
```

```
Unprotect[NonCommutativeMultiply];
NonCommutativeMultiply[x_] := x;
0 ** _ = _ ** 0 = 0;
x_ ** U[] := x; U[] ** x_ := x;
(a_ * x_U) ** (b_ * y_U) := If[a b === 0, 0, Simp[a b (x ** y)]];
(a_ * x_U) ** y_ := Simp[a (x ** y)]; x_ ** (a_ * y_U) := Simp[a (x ** y)];
(x_Plus) ** y_ := (# ** y) & /@ x; x_ ** (y_Plus) := (x ** #) & /@ y;
```

```
U[xx___, x_] ** U[y_, yy___] := If[x ≤ y, U[xx, x, y, yy], U@xx ** (U@y ** U@x + B[U@x, U@y]) ** U@yy];
```

```
UU[l___, x_^n_, r___] := UU[l, Sequence @@ Table[x, {n}], r];
UU[l___, 1, r___] := UU[l, r];
UU[] = U[];
UU[l_, r___] := U[l] ** UU[r];
```

```
UProducts[{}, 0] = {UU[]};
UProducts[{}, n_Integer] /; n > 0 = {};
UProducts[{x_, xs___}, n_Integer] :=
  Sort@Flatten@Table[UU[x^k] ** u, {k, 0, n}, {u, UProducts[{xs}, n - k]}];
UProducts[xs_List, k_Integer, n_Integer] := UProducts[Flatten@Table[x_j, {x, xs}, {j, k}], n];
UProducts[any__, {n_}] := Flatten@Table[UProducts[any, k], {k, 0, n}];
```

```
r_{i_,j_} := Simp[ℏ (h_i UU[1_j] + UU[e_i, f_j])]
```

```
UExp[u_] := Module[{s, t, k},
   s = t = U[]; k = 0;
   While[k < 20 ∧ 0 =!= (t = t ** u), s += t / (++k) !];
   Simp[s]
  ];
R_{i_,j_} := UExp[r_{i,j}];
```

```
UExp[U[e_1]]
```

$$U[] + U[e_1] + \frac{1}{2} U[e_1, e_1] + \frac{1}{6} U[e_1, e_1, e_1] + \frac{1}{24} U[e_1, e_1, e_1, e_1] +$$
$$\frac{1}{120} U[e_1, e_1, e_1, e_1, e_1] + \frac{1}{720} U[e_1, e_1, e_1, e_1, e_1, e_1] + \frac{U[e_1, e_1, e_1, e_1, e_1, e_1, e_1]}{5040} +$$
$$\frac{U[e_1, e_1, e_1, e_1, e_1, e_1, e_1, e_1]}{40320} + \frac{1}{362880} U[e_1, e_1, e_1, e_1, e_1, e_1, e_1, e_1, e_1] +$$
$$U[e_1, e_1, e_1, e_1, e_1, e_1, e_1, e_1, e_1, e_1] / 3628800 + U[e_1, e_1, e_1, e_1, e_1, e_1, e_1, e_1, e_1, e_1, e_1] / 39916800 +$$
$$U[e_1, e_1, e_1, e_1, e_1, e_1, e_1, e_1, e_1, e_1, e_1, e_1] / 479001600 +$$
$$U[e_1, e_1, e_1, e_1, e_1, e_1, e_1, e_1, e_1, e_1, e_1, e_1, e_1] / 6227020800 +$$
$$U[e_1, e_1, e_1, e_1, e_1, e_1, e_1, e_1, e_1, e_1, e_1, e_1, e_1, e_1] / 87178291200 +$$
$$U[e_1, e_1, e_1, e_1, e_1, e_1, e_1, e_1, e_1, e_1, e_1, e_1, e_1, e_1, e_1] / 1307674368000 +$$
$$U[e_1, e_1, e_1, e_1, e_1, e_1, e_1, e_1, e_1, e_1, e_1, e_1, e_1, e_1, e_1, e_1] / 20922789888000 +$$
$$U[e_1, e_1, e_1, e_1, e_1, e_1, e_1, e_1, e_1, e_1, e_1, e_1, e_1, e_1, e_1, e_1, e_1] / 355687428096000 +$$
$$U[e_1, e_1, e_1, e_1, e_1, e_1, e_1, e_1, e_1, e_1, e_1, e_1, e_1, e_1, e_1, e_1, e_1, e_1] / 6402373705728000 +$$
$$U[e_1, e_1, e_1, e_1, e_1, e_1, e_1, e_1, e_1, e_1, e_1, e_1, e_1, e_1, e_1, e_1, e_1, e_1, e_1] / 121645100408832000 +$$
$$U[e_1, e_1, e_1, e_1, e_1, e_1, e_1, e_1, e_1, e_1, e_1, e_1, e_1, e_1, e_1, e_1, e_1, e_1, e_1, e_1] / 2432902008176640000$$

```
UExp[ℏ U[e_1]]
```

$$U[] + \hbar U[e_1] + \frac{1}{2} \hbar^2 U[e_1, e_1] + \frac{1}{6} \hbar^3 U[e_1, e_1, e_1]$$

```
$TD = 5; UExp[ℏ U[e1]]
```

$U[] + \hbar U[e_1] + \frac{1}{2}\hbar^2 U[e_1, e_1] + \frac{1}{6}\hbar^3 U[e_1, e_1, e_1] + \frac{1}{24}\hbar^4 U[e_1, e_1, e_1, e_1] + \frac{1}{120}\hbar^5 U[e_1, e_1, e_1, e_1, e_1]$

```
$TD = 3; Simp[R1,2 ** R1,3 ** R2,3 - R2,3 ** R1,3 ** R1,2]
```

0

```
$TD = 4; Simp[R1,2 ** R1,3 ** R2,3 - R2,3 ** R1,3 ** R1,2]
```

0

```
R1,2
```

$U[] + \hbar h_1 U[l_2] + \left(\hbar + \frac{\hbar^2 h_1}{2} + \frac{1}{6}\hbar^3 h_1^2 + \frac{1}{24}\hbar^4 h_1^3\right) U[e_1, f_2] + \frac{1}{2}\hbar^2 h_1^2 U[l_2, l_2] + \left(\hbar^2 h_1 + \frac{1}{2}\hbar^3 h_1^2 + \frac{1}{6}\hbar^4 h_1^3\right) U[e_1, l_2, f_2] + \frac{1}{6}\hbar^3 h_1^3 U[l_2, l_2, l_2] + \left(\frac{\hbar^2}{2} + \frac{\hbar^3 h_1}{2} + \frac{7}{24}\hbar^4 h_1^2\right) U[e_1, e_1, f_2, f_2] + \left(\frac{1}{2}\hbar^3 h_1^2 + \frac{1}{4}\hbar^4 h_1^3\right) U[e_1, l_2, l_2, f_2] + \frac{1}{24}\hbar^4 h_1^4 U[l_2, l_2, l_2, l_2] + \left(\frac{\hbar^3 h_1}{2} + \frac{1}{2}\hbar^4 h_1^2\right) U[e_1, e_1, l_2, f_2, f_2] + \frac{1}{6}\hbar^4 h_1^3 U[e_1, l_2, l_2, l_2, f_2] + \left(\frac{\hbar^3}{6} + \frac{\hbar^4 h_1}{4}\right) U[e_1, e_1, e_1, f_2, f_2, f_2] + \frac{1}{4}\hbar^4 h_1^2 U[e_1, e_1, l_2, l_2, f_2, f_2] + \frac{1}{6}\hbar^4 h_1 U[e_1, e_1, e_1, l_2, f_2, f_2, f_2] + \frac{1}{24}\hbar^4 U[e_1, e_1, e_1, e_1, f_2, f_2, f_2, f_2]$

old above / new below

The "Internal Multiplication" and Meta-Associativity

```
Join[f[a, b], f[b, c]]
```

f[a, b, b, c]

```
? Cases
```

Cases[{e_1, e_2, …}, *pattern*] gives a list of the e_i that match the pattern.
Cases[{e_1, …}, *pattern* → *rhs*] gives a list of the values of *rhs* corresponding to the e_i that match the pattern.
Cases[*expr*, *pattern*, *levelspec*] gives a list of all parts of *expr* on levels specified by *levelspec* that match the pattern.
Cases[*expr*, *pattern* → *rhs*, *levelspec*] gives the values of *rhs* that match the pattern.
Cases[*expr*, *pattern*, *levelspec*, *n*] gives the first *n* parts in *expr* that match the pattern.
Cases[*pattern*] represents an operator form of Cases that can be applied to an expression. >>

```
Cases[{a1, b2, c1}, x_1]
```

{a1, c1}

```
Cases[{a1, b2, c1}, x_1 :→ f[x]]
```

{f[a], f[c]}

```
m[i_, j_, k_][ℰ_] := Simp[ℰ /. {
    u_U :→ UU @@ Join[
        DeleteCases[u, x_i|j],
        U @@ Cases[u, x_i :→ xk],
        U @@ Cases[u, x_j :→ xk]
      ],
    hi|j → hk
  }]
```

```
UU[e1, l4, f2]
```

U[e1, l4, f2]

```
UU[e1, l4, f2] // m[1, 2, 3]
```

$U[e_3, l_4, f_3]$

```
UU[e1, l4, f2] // m[2, 1, 3]
```

$-h_3 U[l_4] + U[e_3, l_4, f_3]$

```
Union@Table[
   (u // m[2, 1, 1] // m[1, 3, 1]) - (u // m[2, 3, 2] // m[1, 2, 1]),
   {u, UProducts[{e, l, f}, 4, {3}]}
  ] // Short
```

$\{0, -h_1 U[], \ll 55 \gg, -U[l_4, f_1], U[l_4, f_1]\}$

```
Union@Table[
  (u // m[1, 2, 1] // m[1, 3, 1]) - (u // m[2, 3, 2] // m[1, 2, 1]),
  {u, UProducts[{e, l, f}, 4, {3}]}
 ]
```

$\{0\}$

The Invariant of the Trefoil




```
$TD = 2; R4,1 ** R2,5 ** R6,3
```

$$U[] + \hbar h_4 U[l_1] + \hbar h_6 U[l_3] + \hbar h_2 U[l_5] + \left(\hbar + \frac{\hbar^2 h_2}{2}\right) U[e_2, f_5] + \left(\hbar + \frac{\hbar^2 h_4}{2}\right) U[e_4, f_1] + \left(\hbar + \frac{\hbar^2 h_6}{2}\right) U[e_6, f_3] + \frac{1}{2}\hbar^2 h_4^2 U[l_1, l_1] + \hbar^2 h_4 h_6 U[l_1, l_3] + \hbar^2 h_2 h_4 U[l_1, l_5] + \frac{1}{2}\hbar^2 h_6^2 U[l_3, l_3] + \hbar^2 h_2 h_6 U[l_3, l_5] + \frac{1}{2}\hbar^2 h_2^2 U[l_5, l_5] + \hbar^2 h_4 U[e_2, l_1, f_5] + \hbar^2 h_6 U[e_2, l_3, f_5] + \hbar^2 h_2 U[e_2, l_5, f_5] + \hbar^2 h_4 U[e_4, l_1, f_1] + \hbar^2 h_6 U[e_4, l_3, f_1] + \hbar^2 h_2 U[e_4, l_5, f_1] + \hbar^2 h_4 U[e_6, l_1, f_3] + \hbar^2 h_6 U[e_6, l_3, f_3] + \hbar^2 h_2 U[e_6, l_5, f_3] + \frac{1}{2}\hbar^2 U[e_2, e_2, f_5, f_5] + \hbar^2 U[e_2, e_4, f_1, f_5] + \hbar^2 U[e_2, e_6, f_3, f_5] + \frac{1}{2}\hbar^2 U[e_4, e_4, f_1, f_1] + \hbar^2 U[e_4, e_6, f_1, f_3] + \frac{1}{2}\hbar^2 U[e_6, e_6, f_3, f_3]$$

```
$TD = 2; R4,1 ** R2,5 ** R6,3 // m[1, 2, 1] // m[1, 3, 1] // m[1, 4, 1] // m[1, 5, 1] // m[1, 6, 1]
```

$$\left(1 - 2\hbar h_1 + \hbar^2 h_1^2\right) U[] + \left(3\hbar h_1 - 6\hbar^2 h_1^2\right) U[l_1] + \left(3\hbar - \frac{3\hbar^2 h_1}{2}\right) U[e_1, f_1] + \frac{9}{2}\hbar^2 h_1^2 U[l_1, l_1] + 9\hbar^2 h_1 U[e_1, l_1, f_1] + \frac{9}{2}\hbar^2 U[e_1, e_1, f_1, f_1]$$

```
Timing[$TD = 3; R4,1 ** R2,5 ** R6,3 // m[1, 2, 1] // m[1, 3, 1] // m[1, 4, 1] // m[1, 5, 1] // m[1, 6, 1]]
```

$$\Big\{0.546875, \left(1 - 2\hbar h_1 + \hbar^2 h_1^2 + \frac{2}{3}\hbar^3 h_1^3\right) U[] + \left(3\hbar h_1 - 6\hbar^2 h_1^2 + 3\hbar^3 h_1^3\right) U[l_1] + \left(3\hbar - \frac{3\hbar^2 h_1}{2} - \frac{3}{2}\hbar^3 h_1^2\right) U[e_1, f_1] + \left(\frac{9}{2}\hbar^2 h_1^2 - 9\hbar^3 h_1^3\right) U[l_1, l_1] + \left(9\hbar^2 h_1 - \frac{9}{2}\hbar^3 h_1^2\right) U[e_1, l_1, f_1] + \frac{9}{2}\hbar^3 h_1^3 U[l_1, l_1, l_1] + \left(\frac{9\hbar^2}{2} + \frac{9\hbar^3 h_1}{2}\right) U[e_1, e_1, f_1, f_1] + \frac{27}{2}\hbar^3 h_1^2 U[e_1, l_1, l_1, f_1] + \frac{27}{2}\hbar^3 h_1 U[e_1, e_1, l_1, f_1, f_1] + \frac{9}{2}\hbar^3 U[e_1, e_1, e_1, f_1, f_1, f_1]\Big\}$$

```
Timing[$TD = 4; R4,1 ** R2,5 ** R6,3 // m[1, 2, 1] // m[1, 3, 1] // m[1, 4, 1] // m[1, 5, 1] // m[1, 6, 1] // Short]
```

$\{2.75, (\ll 1 \gg) U[] + \ll 14 \gg\}$

Ordering Symbols

```
𝕆[poly_, specs___] := Module[{vs, us, z},
  vs = Join @@ (First /@ {specs});
  us = Join @@ ({specs} /. (l_ → s_) ⧴ (l /. x_i_ ⧴ x_s));
  Simp@Total[CoefficientRules[Normal@Series[poly, {ℏ, 0, $TD}], vs] /. (p_ → c_) ⧴ c UU @@ (us^p)]
 ]
```

Theorem. $R = \mathbb{O}\left(\exp\left(hl + \frac{e^h - 1}{h} ef\right) \mid e \otimes lf\right)$.

Brute Proof.

```
$TD = 6; 𝕆[Exp[ℏ h1 l2 + (e^(ℏ h1) - 1)/h1 e1 f2], {e1} → 1, {l2, f2} → 2] == R1,2
```

```
True
```

```
Timing[$TD = 3;
 𝕆[Exp[ℏ h l1 + (e^(ℏ h) - 1)/h e4 f1 + ℏ h l5 + (e^(ℏ h) - 1)/h e2 f5 + ℏ h l3 + (e^(ℏ h) - 1)/h e6 f3],
   {l1, f1, e2, l3, f3, e4, l5, f5, e6} → 1] /. h1 → h]
```

```
Timing[$TD = 6;
 𝕆[Exp[ℏ h l1 + (e^(ℏ h) - 1)/h e4 f1 + ℏ h l5 + (e^(ℏ h) - 1)/h e2 f5 + ℏ h l3 + (e^(ℏ h) - 1)/h e6 f3],
   {l1, f1, e2, l3, f3, e4, l5, f5, e6} → 1] /. h1 → h]
```

The Big g_0 Lemma.

1. $\mathbb{O}(e^{\gamma l+\beta e} \mid le) = \mathbb{O}(e^{\gamma l+e^\gamma \beta e} \mid el)$.
2. $\mathbb{O}(e^{\gamma l+\beta f} \mid fe) = \mathbb{O}(e^{\gamma l+e^\gamma \beta f} \mid lf)$.
3. $\mathbb{O}(e^{\beta e+\alpha f+\delta ef} \mid fe) = \mathbb{O}(\nu e^{\nu(-\alpha\beta h+\beta e+\alpha f+\delta ef)} \mid ef)$.

Brute Proof.

```
$TD = 3; 𝕆[e^(ℏ γ l1 + β ℏ e1), {l1, e1} → 1]

$TD = 3; 𝕆[e^(ℏ γ l1 + e^(ℏ γ) β ℏ e1), {e1, l1} → 1]

$TD = 6; 𝕆[e^(ℏ γ l1 + β ℏ e1), {l1, e1} → 1] == 𝕆[e^(ℏ γ l1 + e^(ℏ γ) β ℏ e1), {e1, l1} → 1]

$TD = 6; 𝕆[e^(ℏ γ l1 + β ℏ f1), {f1, l1} → 1] == 𝕆[e^(ℏ γ l1 + e^(ℏ γ) β ℏ f1), {l1, f1} → 1]

$TD = 3; 𝕆[e^(ℏ (β e1 + α f1 + δ e1 f1)), {f1, e1} → 1]

$TD = 6; With[{ν = (1 + ℏ h δ)^-1},
 𝕆[e^(ℏ (β e1 + α f1 + δ e1 f1)), {f1, e1} → 1] == 𝕆[ν e^(ℏ ν (-ℏ h α β + β e1 + α f1 + δ e1 f1)), {e1, f1} → 1] /. h1 → h
]
```